University of Camerino

MSc in Computer Science (LM-18)
Performance Analysis and Simulation Course

Winter Wheat Growth Cycle

Student

Roberto Cesetti

Professor

Michele Loreti

A.A. 2022/2023

Table of Contents

Introduction

The growth cycle of winter wheat is defined by distinct phases, each of which is influenced by various factors, including soil nutrient composition, weather conditions, and fertilizer application. Observing and analyzing the growth of a wheat plantation under different conditions, from optimal to difficult, offers a fascinating opportunity to understand the key factors influencing this process.

In order to fully grasp the model and the results it generates, it is essential to possess a solid comprehension of the life cycle of winter wheat and the various factors that influence it.

Stages of Winter Wheat Growth Cycle

The winter wheat growth cycle lasts between 9 and 10 months and here's a general timeline of the growth stages:

1. *Germination*: After planting, the seed will absorb soil's nutrients and begin to swell. The seed coat will then crack open, and a small root will emerge from one end, followed by a small plant that rises above the soil.
2. *Tillering*: The wheat plant will continue to grow leaves and stems, forming a root system that reaches deeper into the soil. The plant will start to produce tillers, which are side shoots that grow from the main stem. At the end of this stage, the seeds that have not germinated will rot. This stage can last for several weeks to a few months, depending on the growing conditions.
3. *Stem Extension*: The wheat plant will continue to grow taller, with the stem elongating at a rapid pace. While more tillers and leaves are produced.
   This stage can last for several weeks to a few months.
4. *Booting and Heading stage*: The wheat plant begins to form spikes, at the top of the Tillers. The spike will be enclosed in a protective sheath, called a boot, by the end of this phase spike will emerge from the protective sheath. These phases can last for several weeks to a few months.
5. *Flowering and Grain Fill*: The wheat plant will enter its reproductive stage, with Flowers appearing on the spikes. Pollination will occur, and the flowers will develop into kernels of wheat. These kernels will fill with starch and protein, resulting in a mature grain. These phases last few months.

6. *Ripening and Harvesting*: The wheat Kernels continue to ripen until they reach maturity. The wheat crop is ready for harvest when the grain is fully mature and the plant has turned golden brown in color. These phases can last from several weeks to months.

Influencing Factors

During the early stages of growth, wheat seeds need adequate soil nutrients to germinate. Under favorable conditions, the plant start emerge within ten days.

Until the first ear emerges, the seedling depends on the energy and nutrients stored in the seed. The plant start developing tillers basing on environmental conditions such as daily water received and temperature. Wheat seeds enjoy an optimal temperature between 10°C and 20°C and 4mm-6mm of water per day.

As the growth cycle progresses, the Tillers begin to produce Flowers, but it is important to know that not all Tillers will flower, in fact at best only half of the Tillers will produce a flower. Once the flowers appear the quantity of fertilizer applied to them and the soil's nutrients will determine how many Kernels there will be per flower, at best a flower produces from 20 to 50 kernels.

Model

The population model developed attempts to represent the growth cycle of winter wheat through its phases, taking into account influencing factors.

Assumptions

The model assumes that:

- The *water* received by the plantation is understood as the average amount of water received per day.
- The *temperature* of the growing environment is understood as the average daily temperature.
- The f*ertilizer* applied to wheat plants means the average amount (in grams) of fertilizer applied to a wheat plant during its life cycle.
- No *seed* will remain after the simulation, which means that each seed that has not produced a plant will rot as it is unable to germinate.

Parameters

The parameters of the model have the responsibility to represent the four influencing factors: *soil_nutrients*, *daily_water*, *temperature* and *fertilizer.*

param soil_nutrients;

It represents the percentage of nutrients in the soil so it is expressed as a value between 0 and 1.

param daily_water;

It represents the average daily water received (in mm) by the plantation during its growth period (best conditions: 4-6mm).

param temperature;

It represents the average daily temperature (in °C) during the growth cycle period (best conditions: 10-20°C).

param fertilizer;

It represents the average amount of fertilizer applied per plant (best conditions: 2g).

Constants

const max_tillers;
It represents the maximum number of tillers that a wheat plant can have in nature.

const max_kernels;
It represents the maximum number of kernels per flower that a wheat plant can have in nature.

const productive_tillers_rate;
It represent the natural capability of a tiller to produce a flower.

const daily_water_rate = 1/(1+(((daily_water-5))^4));
Specifies a function that returns a value representing the probability that a certain rule will be executed based on the value of the param *daily_water*.

const temperature_rate = 1/(1+(((temperature-20)/5)^2));
Specifies a function that returns a value representing the probability that a certain rule will be executed based on the value of the param *temperature*.

const fertilization_rate = 1/(1+(fertilizer-(2/soil_nutrients))^4);
Specifies a function that returns a value representing the probability that a certain rule will be executed based on the value of the params *fertilizer* and *soil_nutrients*.

Species

The species in the model represent the different components that develop during the growth process of the wheat plant.

They are: *S*, *RS*, *P*, *T*, *F* and *K*.

species S;
It represents *seeds* in the model, the species from which the winter wheat growth cycle begins.

species RS;
It represents *rotten seeds* in the model, seeds that do not develop in a plant, due to lack of nutrients, rot.

species P of [0, max_tillers];
It represents *plants* of wheat in the model, the seeds that germinate will produce a plant that then grows.

species T of [0, max_flowers];
It represents *tillers* in the model, they are distinct small branches of the plant and their responsibility is to develop flowers.

species F of [0, max_kernels];
It represents *flowers* in the model, they are produced from tillers and have the responsibility of containing kernels of wheat.

species K;
It represents *kernels* in the model, they are the final results of the growth cycle and, therefore, represent the yield of the plantation.

Rules

The model's rules represent how the growth cycle proceeds, in particular, they define how the transition from one state to another occurs in the plant growth process.

```
rule seed_produces_plant {
   S -[(soil_nutrients)]-> P[max_tillers-1]
}
```

It determines whether a *seed* becomes a *plant* based on the presence of nutrients in the soil.

```
rule seed_rots {
   S -[(1-(soil_nutrients))]-> RS
}
```

It determines whether a *seed* becomes a *rotten seed* based on the presence of nutrients in the soil.

```
rule plants_produce_tillers for i in [1, max_tillers] {
   P[i] -[environmental_conditions]-> T[max_flowers-1] | P[i-1]
}
```

It determines whether a *plant* can produce *tillers* based on environmental conditions such as *daily_water_rate* and *temperature_rate.*

rule tillers_produce_flowers **for** i **in** [1, max_flowers] {
   T[i] -[soil_nutrients * environmental_conditions * productive_tillers_rate]->
         F[max_kernels-1] | T[i-1]
}
It determines whether a *tiller* can produce a *flower* based on environmental conditions, such as *daily_water_rate* and *temperature_rate*, levels of nutrients in the soil and the tillers production rate.

rule flowers_produce_kernels **for** i **in** [1, max_kernels] {
   F[i] -[fertilization_rate * environmental_conditions]-> K | F[i-1]
}
It determines whether a *flower* can produce *kernels* based on environmental conditions, such as *daily_water_rate* and *temperature_rate*, and the *fertilization_rate* which is determined by the value of fertilizer applied to the plant and the soil's nutrients.

Measures

Model measures specify which aspects of the model are taken into account and consulted at the end of the simulation.

measure seeds;
It specifies the number of seeds during the growth cycle.

measure rotten_seeds;
It specifies the number of rotten seeds during the growth cycle.

measure plants;
It specifies the number of plants during the growth cycle.

measure tillers;
It specifies the number of tillers during the growth cycle.

measure flowers;
It specifies the number of flowers during the growth cycle.

measure kernels;
It specifies the number of kernels during the growth cycle.

Initial states

The model’s initial state for the simulation is:

system init = S<100>;
It means that at the beginning of the simulation there are 100 copies of the species S.

Code

Population Model

```
/*──────────────── Winter Wheat Growth Cycle ────────────────*/

/* _-_-_-_-_-_-_-_-_-_-_ Params _-_-_-_-_-_-_-_-_-_-_ */

/* The percentage of nutrients in the soil (between 0 and 1) */
param soil_nutrients = 1;

/* Average daily water (mm) received by the plantation during its
growth period (best conditions: 4–6mm) */
param daily_water = 5;

/* Average daily temperature (°C) during the growth cycle period
(best conditions: 10–20°C) */
param temperature = 15;

/* Represents the amount of fertilizer applied per plant (best
conditions: 2g) */
param fertilizer = 2;

/* _-_-_-_-_-_-_-_-_-_-_ Consts _-_-_-_-_-_-_-_-_-_-_ */

const max_tillers = 6;

const max_flowers = 2;

const max_kernels = 23;

const productive_tillers_rate = 0.5;

const daily_water_rate = 1/(1+(((daily_water-5))^4));

const temperature_rate = 1/(1+(((temperature-20)/5)^2));

const environmental_conditions = daily_water_rate *
temperature_rate;

const fertilization_rate = 1/(1+(fertilizer-(2/
soil_nutrients))^4);
```

```
/* _-_-_-_-_-_-_-_-_-_-_ Species _-_-_-_-_-_-_-_-_-_-_ */


/* Seeds */
species S;

/* Rotten seeds */
species RS;

/* Plants */
species P of [0, max_tillers];

/* Tillers */
species T of [0, max_flowers];

/* Florets */
species F of [0, max_kernels];

/* Kernels */
species K;


/* _-_-_-_-_-_-_-_-_-_-_ Rules _-_-_-_-_-_-_-_-_-_-_ */


/* The Seed produces Plant depending on soil nutrients */
rule seed_produces_plant {
    S -[(soil_nutrients)]-> P[max_tillers-1]
}

/* A seed is considered rotten if it does not have enough
nutrients to produce Plants */
rule seed_rots {
    S -[(1-(soil_nutrients))]-> RS
}

/* Plants produce Tillers depending on environmental conditions */
rule plants_produce_tillers for i in [1, max_tillers] {
    P[i] -[environmental_conditions]-> T[max_flowers-1] | P[i-1]
}

/* Tillers produce Flowers */
rule tillers_produce_flowers for i in [1, max_flowers] {
    T[i] -[soil_nutrients * environmental_conditions *
productive_tillers_rate]-> F[max_kernels-1] | T[i-1]
}

/* Flowers produce Kernels */
rule flowers_produce_kernels for i in [1, max_kernels] {
    F[i] -[fertilization_rate * environmental_conditions]-> K |
F[i-1]
}
```

```
/* _-_-_-_-_-_-_-_-_-_-_ Measures _-_-_-_-_-_-_-_-_-_-_ */


/* The number of seeds during the growth cycle */
measure seeds = #S;

/* The number of rotten seeds during the growth cycle */
measure rotten_seeds = #RS;

/* The number of plants during the growth cycle */
measure plants = #P[i for i in [0, max_tillers]];

/* The number of tillers during the growth cycle */
measure tillers = #T[i for i in [0, max_flowers]];

/* The number of flowers during the growth cycle */
measure flowers = #F[i for i in [0, max_kernels]];

/* The number of kernels during the growth cycle */
measure kernels = #K;


/* _-_-_-_-_-_-_-_-_-_-_ Initial States _-_-_-_-_-_-_-_-_-_-_ */


system init = S<100>;

/* ------------------------------------------------------------------*/
```

Sibilla Shell Script

```
module "population"
load "winter-wheat-growth-cycle.pm"
add all measures
deadline 1000
dt 1.0
replica 10
init "init"
simulate
save output "./results" prefix "wheat_" postfix “__"
```

Simulations

Every simulation serves as a case study that focuses on a specific scenario within the winter wheat growth cycle. Ultimately, these simulations are compared to gain valuable insights into the distinct behaviors exhibited by the model.

The Sibilla configuration, written in the Sibilla Shell Script section, was employed to conduct the following simulations. Specifically, these simulations were carried out over 1000 time units, with a delta time of 1 unit, and replicated 10 times to ensure a more comprehensive and generalized analysis.

For each simulation, a line graph was generated to illustrate the trends of the six measurements outlined in the model:

- Number of Seeds
- Number of Rotten Seeds
- Number of Tillers
- Number of Flowers
- Number of Kernels

These graphs not only depict the average values of these measurements across 10 replications but also showcase the variance and standard deviation associated with each measurement.

Case Study 1: Best Overall Conditions

This initial case study provides valuable insights into the optimal behavior of the model when all parameters are set to represent the ideal conditions for the winter wheat growth cycle.

Params

soil_nutrients = 1;

daily_water = 5;

temperature = 15;

fertilizer = 2;

Initial State

system init = S<100>;

Simulations













Analysis

Simulation results under these specific conditions reveal that the initial 100 seeds successfully develop into plants due to the high percentage of soil nutrients (set to 1). Furthermore, it is worth noting that there are no cases of rotten seeds due to the favorable levels of nutrients in the soil.

Thanks to the optimal environmental conditions, the development of tillers from each plant is maximized, with each plant producing the maximum number of 5 tillers, which is typically the highest achievable number in nature.

Under these environmental conditions and with the assistance of nutrient-rich soil, the tillers are able to produce a considerable number of flowers, highlighting the positive influence of the environmental factors and soil nutrients on the reproductive capacity of the tillers.

Finally, kernels production also benefits from environmental conditions, soil and fertilizers used in appropriate doses, and indeed production is at its peak.

Case Study 2: Adverse Weather Conditions

This case study provides insight into the behavior of the model when environmental conditions, i.e. daily water and temperature, are not favorable for the development of the grain growth process.

Params

soil_nutrients = 0.8;

daily_water = 3;

temperature = 10;

fertilizer = 2;

Initial State

system init = S<100>;

Simulations



Analysis

The simulation highlights that a majority of the seeds successfully germinate into plants due to the presence of acceptable soil nutrient levels, which helps prevent the occurrence of rotten seeds. However, the scarcity of water and low temperature negatively impact the production of tillers. On average, only one tiller is produced for every two plants, and not all tillers go on to flower. Consequently, the average number of flowers is approximately 4, leading to an average yield of only around 18 kernels. This indicates a significant loss in yield under these conditions, showcasing the adverse effects of limited water and low temperature on crop productivity.

Case Study 3: Dry Weather Conditions

This case study aims to highlights the susceptibility of winter wheat to drought conditions, emphasizing the importance of adequate water supply and moderate temperatures for optimal crop growth and productivity.

Params

soil_nutrients = 0.4;

daily_water = 2.5;

temperature = 30;

fertilizer = 2;

Initial State

system init = S<100>;

Simulations

Analysis

The simulation results reveal the impact of these unfavorable environmental factors on the growth cycle of winter wheat. Due to limited water, the germination rate of seeds is significantly reduced, resulting in a lower number of plants compared to optimal conditions. Additionally, the high temperature negatively affects the development of tillers, leading to a reduced number of tillers produced by each plant. Furthermore, the scarcity of water and high temperature adversely affect the flowering stage, resulting in a lower overall number of flowers. As a result, the number of kernels produced is zero, with a significant loss in yield.

Case Study 4: Excessively Wet Weather

This case study aims to highlights the behavior of the model in wet weather conditions showing how even too much water can negatively affect the growth process.

Params

soil_nutrients = 0.4;

daily_water = 7;

temperature = 8;

fertilizer = 2;

Initial State

system init = S<100>;

Simulations











Analysis

The simulation results reveal the impact of excessive water on the growth cycle of winter wheat. The excessive moisture saturates the soil, creating an environment conducive to seed rot instead of providing the necessary conditions for germination. The surplus water inhibits the production of tillers, resulting in a lower number of tillers per plant compared to optimal conditions. Additionally, the excessive moisture negatively affects the flowering stage, leading to a reduced number of flowers per tiller.

Consequently, the number of kernels produced is significantly diminished, resulting in a lower yield compared to ideal conditions.

Case Study 5: Low Soil Nutrients

This case study aims to highlights the crucial role played by soil nutrients in influencing the yield of winter wheat crops. The simulation results show that seed growth and development strongly depend on the availability of nutrients in the soil despite optimal atmospheric conditions and fertilizer levels.

Params

soil_nutrients = 0.4;

daily_water = 5;

temperature = 15;

fertilizer = 2;

Initial State

system init = S<100>;

Simulations



Analysis

The simulation highlights the significant impact of soil nutrients on the growth cycle of winter wheat, planting winter wheat in soil lacking sufficient nutrients results in plants that can produce tillers and flowers, but the overall yield at the end of the process is significantly lower than in optimal conditions due to the fact that a lot of seeds aren't able to produce plants. This case study underscores the importance of providing adequate soil fertility and nutrient levels to ensure successful winter wheat cultivation and improve crop growth and yield.

Case Study 6: Not Enough Fertilizer

The study highlights the importance of proper fertilization practices in winter wheat cultivation, which can significantly improve crop growth and yield. It underscores the need for growers to carefully manage and monitor fertilizer application.

Params

soil_nutrients = 0.8;

daily_water = 5;

temperature = 15;

fertilizer = 0.2;

Initial State

system init = S<100>;

Simulations



Analysis

The simulations provide valuable insights into how fertilizer influences the production of kernels in winter wheat. The growth process of wheat proceeds smoothly until the flowering stage, thanks to optimal weather conditions and the presence of sufficient soil nutrients. However, the number of kernels produced is strongly influenced by the quantity of fertilizer applied.
It becomes evident that an inadequate amount of fertilizer results in a lower yield compared to optimal overall conditions. The presence of optimal weather conditions and soil nutrients alone is not sufficient to maximize kernel production.

Case Study 7: Too Much Fertilizer

The study underscores the significance of employing appropriate fertilization practices in winter wheat cultivation, as they play a crucial role in enhancing crop growth and maximizing yield. However, it is important to note that the application of excessive fertilizer can have adverse effects and result in a loss in wheat yield.

Params

soil_nutrients = 0.8;

daily_water = 5;

temperature = 15;

fertilizer = 5;

Initial State

system init = S<100>;

Simulations



Analysis

This simulations prove that the over-application of certain nutrients, such as nitrogen, can result in rapid plant growth but with negative effects on kernel development, indeed the simulation shows that more tillers are produced compared to optimal conditions but their quality is compromised, resulting in a lower final number of kernels. This case study highlights the importance of precision in fertilizer application, ensuring that the nutrient levels are properly balanced to promote healthy plant growth and optimize kernel production.

Conclusions

The simulations conducted on the growth cycle of winter wheat provide valuable insights into the various conditions and factors that influence crop productivity.

Under optimal conditions, including favorable weather and proper fertilizer application, the wheat plantation experiences robust growth and achieves high yields. The presence of adequate soil nutrients promotes successful germination and healthy plant development, resulting in abundant tillers, flowers, and kernels.

Conversely, planting winter wheat in soil lacking sufficient nutrients or subjecting it to drought and high temperatures leads to diminished growth and reduced yields. Insufficient soil nutrients impede plant development and result in fewer kernels, while water scarcity and extreme heat hinder both germination and overall plant growth.

Interestingly, excessive water availability can also have adverse effects on wheat growth, leading to seed rot and compromised plant development. Similarly, excessive fertilizer application, particularly with nitrogen, can trigger rapid but weak plant growth, resulting in diminished kernel production and lower quality overall.

In conclusion, the simulations presented in this study provide valuable insights into the growth cycle of winter wheat under different conditions. However, it is important to recognize that these simulations represent only a limited range of possible scenarios. To increase the accuracy and completeness of the model, further improvements can be made by incorporating additional influencing factors, such as pests, diseases, and environmental variables. Furthermore, refining the model with additional rules that better represent the complexities of wheat growth will result in more accurate simulations.
These advances will contribute to a deeper understanding of winter wheat growth cycle.

References

Articles

Growth Stages of Wheat:
Identification and Understanding Improve Crop Management
By Travis D. Miller
Link: http://lubbock.tamu.edu/files/2011/10/wheatgrowthstages_26.pdf

The Wheat Plant and Its Life Cycle
Link: https://www.reacchpna.org/sites/default/files/Elem%20Curriculum%20W1-D4.pdf

Wheat growth and physiology
By E. Acevedo, P. Silva, H. Silva
Link: https://www.fao.org/3/y4011e/y4011e06.htm

The Winter Wheat Life Cycle:
From Seedling to Supermarket
Link: https://blog.machinefinder.com/15099/the-winter-wheat-life-cycle-from-seedling-to-supermarket